NAG Library Routine Document

F03BFF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F03BFF computes the determinant of a real n by n symmetric positive definite matrix A. F07FDF (DPOTRF) must be called first to supply the symmetric matrix A in Cholesky factorized form. The storage (upper or lower triangular) used by F07FDF (DPOTRF) is not relevant to F03BFF since only the diagonal elements of the factorized A are referenced.

2 Specification

```
SUBROUTINE F03BFF (N, A, LDA, D, ID, IFAIL)
INTEGER            N, LDA, ID, IFAIL
REAL (KIND=nag_wp) A(LDA,*), D
```

3 Description

F03BFF computes the determinant of a real n by n symmetric positive definite matrix A that has been factorized as $A = U^{\mathrm{T}}U$, where U is upper triangular, or $A = LL^{\mathrm{T}}$, where L is lower triangular. The determinant is the product of the squares of the diagonal elements of U or L. The Cholesky factorized form of the matrix must be supplied; this is returned by a call to F07FDF (DPOTRF).

4 References

Wilkinson J H and Reinsch C (1971) *Handbook for Automatic Computation II, Linear Algebra* Springer–Verlag

5 Arguments

1: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N > 0.

2: A(LDA, $*$) – REAL (KIND=nag_wp) array *Input*

Note: the second dimension of the array A must be at least N.

On entry: the lower or upper triangle of the Cholesky factorized form of the n by n positive definite symmetric matrix A. Only the diagonal elements are referenced.

3: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F03BFF is called.

Constraint: LDA $\geq$ N.

4: D – REAL (KIND=nag_wp) *Output*

5: ID – INTEGER *Output*

On exit: the determinant of A is given by D $\times$ 2.0^{ID}. It is given in this form to avoid overflow or underflow.

6: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, N $= \langle value \rangle$.
Constraint: N > 0.

IFAIL $= 3$

On entry, LDA $= \langle value \rangle$ and N $= \langle value \rangle$.
Constraint: LDA $\geq$ N.

IFAIL $= 4$

The matrix A is not positive definite.

IFAIL $= -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The accuracy of the determinant depends on the conditioning of the original matrix. For a detailed error analysis see page 25 of Wilkinson and Reinsch (1971).

8 Parallelism and Performance

F03BFF is not threaded in any implementation.

9 Further Comments

The time taken by F03BFF is approximately proportional to n.

10 Example

This example computes a Cholesky factorization and calculates the determinant of the real symmetric positive definite matrix

$$\begin{pmatrix} 6 & 7 & 6 & 5 \\ 7 & 11 & 8 & 7 \\ 6 & 8 & 11 & 9 \\ 5 & 7 & 9 & 11 \end{pmatrix}.$$

10.1 Program Text

```
    Program f03bffe

!     F03BFF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: dpotrf, f03bff, nag_wp, x04caf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
      Character (1), Parameter         :: uplo = 'l'
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: d
      Integer                          :: i, id, ifail, info, lda, n
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: a(:,:)
!     .. Executable Statements ..
      Write (nout,*) 'F03BFF Example Program Results'

!     Skip heading in data file
      Read (nin,*)

      Read (nin,*) n
      lda = n
      Allocate (a(lda,n))

      Read (nin,*)(a(i,1:n),i=1,n)

!     Factorize A
!     The NAG name equivalent of dpotrf is f07fdf
      Call dpotrf(uplo,n,a,lda,info)

      Write (nout,*)
      Flush (nout)
      ifail = 0
      Call x04caf('L','N',n,n,a,lda,'Array A after factorization',ifail)

      ifail = 0
      Call f03bff(n,a,lda,d,id,ifail)

      Write (nout,*)
      Write (nout,99999) d, id
      Write (nout,*)
      Write (nout,99998) d*2.0_nag_wp**id

99999 Format (1X,'D = ',F13.5,' ID = ',I12)
99998 Format (1X,'Value of determinant = ',E13.5)
    End Program f03bffe
```

10.2 Program Data

```
F03BFF Example Program Data
  4                         : N
    6    7    6    5
    7   11    8    7
    6    8   11    9
    5    7    9   11    : A
```

10.3 Program Results

```
 F03BFF Example Program Results

 Array A after factorization
             1          2          3          4
 1      2.4495
 2      2.8577     1.6833
 3      2.4495     0.5941     2.1557
 4      2.0412     0.6931     1.6645     1.8927

 D =       0.06909 ID =           12

 Value of determinant =   0.28300E+03
```